
ABSTRACT
One of the unique features of the REPORT procedure is the Compute Block. Unlike most other SAS procedures, PROC
REPORT has the ability to modify values within a column, to insert lines of text into the report, to create columns, and to
control the content of a column. Through compute blocks it is possible to use a number of SAS language elements, many
of which can otherwise only be used in the DATA step.
While powerful, the compute block can also be complex and potentially confusing. This tutorial introduces basic compute
block concepts, statements, and usages. It discusses a few of the issues that tend to cause folks consternation when first
learning how to use the compute block in PROC REPORT.
This paper is being presented in conjunction with the Tutorial PROC REPORT: Compute Block Basics –Part I Tutorial
(Paper 031-2008). Consult that paper for additional details

INTRODUCTION
While the topic of compute blocks in the REPORT step is not particularly difficult or advanced, the discussion of compute
blocks must necessarily assume that the reader has a basic understanding of the REPORT step itself. In the discussion
that follows, an understanding of the COLUMN, DEFINE, BREAK, and RBREAK statements and their relationships to each
other is assumed. The reader should also be aware of the different define usages of report items.
There are two basic types of compute blocks; those that are associated with a location (the option BEFORE or AFTER
follows the COMPUTE keyword), and those associated only with a report item. While the structure and execution of these
two types of compute blocks is similar, how they are used and the timing of their execution can be quite different.
The compute block starts with the COMPUTE statement and terminates with the ENDCOMP statement. Usually the
compute block is placed in the REPORT step after the DEFINE statements. The syntax of the compute block looks
something like:
compute <location> <report_item> </ options>;
one or more SAS language elements
endcomp;
The components of the COMPUTE statement include:
•

location

Specifies when the compute block is to execute and ultimately what is to be done with the result of
the compute block. Accepted values include BEFORE and AFTER. When a location is specified
without also specifying a report_item, the location will be at the start (BEFORE) or at the end
(AFTER) of the report.

•

report_item

When the result of the compute block is associated with a variable or report item, its name is
supplied here. This report_item variable can be any variable on the COLUMN statement. When
report_item is a variable that either groups or orders rows (usage of GROUP or ORDER) you may
also use BEFORE and AFTER to apply the result at the start or end of each group.

•

options

Several options are available that can be used to determine the appearance and location of the
result of the compute block.

•

SAS language elements
Any number of SAS language elements can be used within the compute block. These include the
use of executable statements, logical processing (IF-THEN/ELSE), and most of the functions
available in the DATA step.

The compute block can be placed anywhere within the REPORT step, however generally compute blocks are grouped
following the DEFINE statements.

Learning Through Exercises
During the Hands-on Workshop associated with this paper, a series of exercises were tackled by the students, and these
are included in this paper. Because of the nature of Hands-on Workshops this paper may seem a bit terse. I recommend
that you take the following steps if you want to maximize your learning experience through the use of this paper.
1) First read the companion paper on the same subject in the Tutorial Section (Paper 031-2008).
2) Go through the exercises in this paper (referring back to Paper 031-2008 for details when needed).
Each exercise contains code that requires you to fill in the blanks.
Try to fill in the blanks (represented by xxxxxx and yyyyyyy) and then check your answers.
3) Buy a copy of Carpenter’s Complete Guide to the SAS REPORT Procedure for even more detail.(

USING THE LINE STATEMENT
One of the more interesting programming statements within the compute block is the LINE statement. This statement is
roughly analogous to the PUT statement in the DATA step and can be used to introduce lines of text into the report.

Exercise #1: Inserting a Blank Line
In this exercise REGION is a grouping variable, and we would like to insert a blank line using the LINE statement after
each set of rows for each region. Change the xxxxxxxx’s to insert the blank lines.
************************************************************
* Exercise 1 ;
* Create a blank line after each region;
title1 'Exercise 1';
proc report data=sasclass.clinics nowd;
column region sex wt;
define region / group format=$6.;
define sex
/ group format=$6. 'Gender';
define wt
/ analysis;
compute after xxx;
line xxxxx;
endcomp;
run;

The following REPORT step shows one possible solution to Exercise #1. In this example a blank line has been inserted
AFTER each level of the grouping variable (REGION).
* Blank Line using COMPUTE;
title1 'Using Proc REPORT';
title2 'Blank Line After Region';
proc report data=rptdata.clinics
(where=(region in('1' '2' '3' '4')))
nowd;
column region sex wt,(n mean);
define region / group format=$6.;
define sex
/ group format=$6. 'Gender';

define wt
/ analysis;
compute after region;
line ' ';
endcomp;
run;
Using Proc REPORT
Blank Line After Region
weight in pounds
n
mean
4
195

region
1

Gender
M

2

F
M

6
4

109.66667
105

3

F
M

5
5

127.8
163.8

4

F
M

4
10

143
165.6

The LINE statement has
been used to insert a blank
line AFTER each regional
group.

Because the COMPUTE statement contains the location specification AFTER and also designates the report item
REGION, we have effectively requested that the results of the compute block (a blank space) be written after each group
of the grouping variable REGION.

Exercise #2: Adding Lines of Text
The LINE statement can also be used to add lines of text either at the top or bottom of the report. In Exercise #2 use the
LINE statement to add a title to the report.
************************************************************
* Exercise 2;
* Add text before the report;
title1 'Exercise 2';
proc report data=sasclass.clinics nowd;
column region sex wt;
define region / group format=$6.;
define sex
/ group format=$6. 'Gender';
define wt
/ analysis;
compute xxxxxx;
line xxxxxx;
endcomp;
run;
In the previous example the LINE statement added a blank line, the LINE statement is more commonly used to add lines of
text at specific locations in the report.
In the following example the text is written as a footnote at the end of the report. We can specify the options BEFORE and
AFTER to indicate the location, and since in this case no grouping variable appears on the COMPUTE statement, the
AFTER applies to the whole REPORT.
* Text Line using COMPUTE;
title1 'Using Proc REPORT';
title2 'Footnote Using LINE';
proc report data=rptdata.clinics
(where=(region in('1' '2' '3' '4')))
nowd;
column region sex wt,(n mean);

define region / group format=$6.;
define sex
/ group format=$6. 'Gender';
define wt
/ analysis;
compute after region;
line ' ';
endcomp;
compute after;
line @20 'Weight taken during';
line @20 'the entrance exam.';
endcomp;
run;
Using Proc REPORT
Footnote Using LINE
weight in pounds
n
mean
4
195

region
1

Gender
M

2

F
M

6
4

109.66667
105

3

F
M

5
5

127.8
163.8

4

F
M

4
10

143
165.6

This report was generated
with the system option
NOCENTER in effect
(notice the titles). Even so
the results of the LINE
statement will be centered
unless a column
specification is provided.

Weight taken during
the entrance exam.

In these LINE statements the @ is used, as it is in the DATA step PUT statement, to designate the column number. If a
specific column is not specified with the @, and no justification options are specified, text generated by the LINE statement
will be centered. This is a different default behavior than the PUT statement in the DATA step.
When writing to ODS destinations other than LISTING, proportional fonts may make exact placement of values difficult,
and may require you to use a trial-and-error approach, and to make things more interesting some destinations ignore the
@ altogether.

USING SAS LANGUAGE ELEMENTS
Much of the power of the DATA step is available within the compute block. This radically increases the flexibility of the
REPORT step as most of the SAS language elements, such as, routines, functions, arithmetic operations, and executable
statements can also be used within the compute block. These include DO loops, assignment and sum statements,
ARRAYs, and IF-THEN/ELSE processing.
Most of the functions and routines that do not work in the compute block are those that utilize either the DATA step’s
Program Data Vector, PDV, or the processing of the DATA step itself. Examples include the LAG function and the
RETAIN statement. As was already discussed, the LINE statement replaces the PUT.
The use of %INCLUDE statements, macro variables, and macro invocations work the same in compute blocks as they do
in other parts of SAS. In the case of %INCLUDE statements and elements of the macro language, it is less a matter of
these items working in a compute block, and is more accurately simply a property of Base SAS and how these elements
apply to all procedures.

CREATING AND MODIFYING COLUMNS
One of the great strengths of PROC REPORT is its ability to create columns based on calculations carried out during the

execution of the procedure. Columns that are not on the incoming data set can be created and displayed within the PROC
REPORT step. Often this can eliminate one or more DATA steps. The compute block provides the power and flexibility to
mold the column in a variety of ways.
Compute blocks can be used to create both numeric and character variables, and an extensive number of the SAS
language elements that give the DATA step much of its power and functionality are also available for use in the compute
block.

Coordinating with the COLUMN and DEFINE Statements
Variables on the data set that is to be processed by PROC REPORT are named on the COLUMN statement and the way
that these variables are to be used is specified on the DEFINE statement. This is also true for report item columns
created through the use of the compute block. In addition, the name of the computed variable will also appear on the
COMPUTE statement.
When a column is created through a compute block, the new column is named and that column name is used on both the
COLUMN statement and on a DEFINE statement. Since a new column with have a define usage of COMPUTE, it cannot
also be a grouping or order variable, therefore the BEFORE/AFTER locations cannot be specified on the COMPUTE
statement. Since the name of the computed report item will appear on the COLUMN, DEFINE, and COMPUTE
statements, it is this name that is used to create and coordinate the link between the three statements.
In the following example the patients weights are to be displayed in both pounds and kilograms (the earlier example
converts to kilograms without creating a new column). Since the units for the variable WT are in pounds, a new column
containing the weight in kilograms will need to be created. This is accomplished in a compute block which creates the
computed variable WTKG.
* Creating a new column with a compute block;
title1 'Using The COMPUTE Block';
title2 'Adding a Computed Column';
proc report data=rptdata.clinics nowd split='*';
column lname sex (' Weight *--' wt wtkg); ì
define lname
/ order
width=18 'Last Name*--';
define sex
/ display width=6 'Gender*--';
define wt
/ display format=6. 'Pounds*--';
define wtkg í / computed î format=9.2 'Kilograms*--';
compute wtkg; ï
wtkg = wt / 2.2; ð
endcomp; ñ
run;

ì The name (WTKG) for the new column is added
to the COLUMN statement.
í The DEFINE statement associated with the new
column (WTKG) has a define type of COMPUTED
î.
ï The COMPUTE statement contains the name of
the computed variable.
ð The new variable (WTKG) is calculated by
dividing the weight (WT) in pounds by 2.2.
ñ Compute blocks are terminated with an
ENDCOMP statement.
The resulting table shows both the weight in
pounds and kilograms.

Using The COMPUTE Block
Adding a Computed Column
Weight
ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
Last Name
Gender Pounds Kilograms
ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ ƒƒƒƒƒƒ ƒƒƒƒƒƒ ƒƒƒƒƒƒƒƒƒ
Adams
F
155
70.45
Adamson
F
158
71.82
Alexander
M
175
79.55
Antler
M
240
109.09
Atwood
M
105
47.73
Banner
M
175
79.55
Baron
M
160
72.73
Batell
F
155
70.45
. . . . portions of the table not shown . . . .

Exercise #3: Creating a New Column
The current patient height (HT) is measured in inches. Use the compute block to create a new report column for height in
feet.
************************************************************
* Exercise 3;
* Create a column for height in feet;
title1 'Exercise 3';
proc report data=sasclass.clinics nowd;
column lname fname ht xxxxx;
define lname / order 'Last';
define fname / order 'First';
define ht
/ display 'Inches';
define feet / xxxxx 'Feet';
compute xxxxx;
xxxxx = ht/12;
endcomp;
run;
New columns can be created through the compute block. Any column that is to appear in the report must also appear on
the COLUMN statement and new columns must also have a DEFINE statement with a usage of COMPUTED.
title1 'Using PROC REPORT';
title2 'Creating a Computed Column';
proc report data=sasclass.clinics nowd;
column lname fname ('Height' ht feet);
define lname / order 'Last';
define fname / order 'First';
define ht
/ display 'Inches';
define feet / computed 'Feet' f=4.1;
compute feet;
feet = ht/12;
endcomp;
run;
Using PROC REPORT
Creating a Computed Column
Last
Maxwell
Mercy
Moon
Most
Nabers
Nolan
Olsen
Panda
Perez
Pope
Reilly
Robertson
Rodgers
Ronson

First
Linda
Ronald
Rachel
Mat
David
Terrie
June
Merv
Mathew
Robert
Arthur
Adam
Carl
Gerald

Height
Inches Feet
64
5.3
74
6.2
62
5.2
67
5.6
74
6.2
63
5.3
72
6.0
74
6.2
68
5.7
72
6.0
69
5.8
69
5.8
72
6.0
69
5.8

. . . . . portions of the table not shown . . . . .

Assignment statements have the same form in the compute block as they do in the DATA step.

Exercise #4: Changing the Value of an Existing Column
The value of the patient’s weight (WT) is measured in pounds. Convert this weight to kilograms.
************************************************************
* Exercise 4;
* Convert mean WT from pounds to kilos;
title1 'Exercise 4';
proc report data=sasclass.clinics nowd;
column region n ('Weight' wt);
define region / group
f=$6.;
define n
/ f=3.;
define wt
/ analysis mean 'Kilograms' f=9.2;
compute xxxxx;
wtxxxxx = wtxxxxx / 2.2;
endcomp;
run;
This exercise, along with Exercise #3, demonstrates some of the power of the compute block. This compute block
performs a transformation of weight from pounds to Kilograms. The conversion is done in a compute block with the same
name as the variable (report item) that is being modified. The assignment statement in the compute block is of the same
form as you would expect to find in the DATA step.
title1 'Using Proc REPORT';
title2 'Converting Weight to Kg';
proc report data=rptdata.clinics(where=(region in('4')))
nowd;
column lname fname sex wt;
define lname / display;
define fname / display;
define sex
/ display format=$6. 'Gender';
define wt
/ display 'Weight in Kg'
format=6.2;
compute wt;
* Convert pounds to KG;
wt = wt/2.2;
endcomp;
run;
Using Proc REPORT
Converting Weight to Kg
last name
Halfner
Johnson
Rodgers
Cordoba
Baron
Adams
Rymes
Most
Jackson
Maxim
Perez
East
Batell
Rumor

first
name
John
Randal
Carl
Juan
Roger
Mary
Carol
Mat
Ted
Kurt
Mathew
Clint
Mary
Stacy

Gender
M
M
M
M
M
F
F
M
M
M
M
M
F
F

The label for the variable
WT contains the units
(pounds), so a new
column header is also
required.

Weight
in Kg
70.45
91.36
81.36
60.45
72.73
70.45
59.55
70.45
91.36
81.36
60.45
72.73
70.45
59.55

REPORT ITEM NAMING RULES
When coding within the compute block, report items and temporary variables will need to be addressed. How they are
addressed or named in the compute block depends on what the report item is and how it is being used. There are four
different ways of naming report items within a compute block.

Explicitly by Name
While you can have a compute block for any report item (variable on the COLUMN statement), the naming conventions of
the report items that are addressed within the compute block are not nearly as straight forward as they would seem to be in
the previous two examples. These are about the simplest cases in which the name of the column is used directly
(explicitly).
The variable name can be used directly when the variable has a define type of GROUP, ORDER, COMPUTED, or
DISPLAY. Temporary variables, variables that are created and only used in a compute block, are also always addressed
directly by variable name.
The REPORT step also creates an automatic temporary variable named _BREAK_. This variable is also addressed
directly and is available during the execution of the compute block.

Using a Compound Name
Compound variable names are needed, when a variable has a define usage of ANALYSIS. Whether requested or not, a
statistic is always calculated for analysis variables. The compound name is a combination of the variable name and the
statistic that it has been used to calculate. The general form is variablename.statistic, and an example of a compound
name might be:
wt.mean
When a statistic is not otherwise specified it is assumed to be SUM.

By Specifying an Alias
An alias can be specified in the COLUMN statement when you want to use a single analysis variable in more than one way
- generally to calculate more than one statistic. When used in a compute block, aliases are named explicitly. The following
COLUMN statement generates a series of aliases for the HT analysis variable.
columns region ht ht=htmin ht=htmax
ht=htmean ht=htmedian;
In the compute block the alias is addressed directly as in:
compute after;
line @3 'Minimum height is ' htmin 6.1;
endcomp;

Indirectly Using the Absolute Column Number
Sometimes as the report is constructed a given column may not have a specific name. This is especially the case when a
variable, with the define type of ACROSS, creates a series of columns. These, and indeed any column in the report, can
be referenced by using the absolute column number as an indirect column reference. This column number is included in a
pseudo variable name which is always of the form
_Cxx_
where the xx is the column number as read from left to right on the report. Keep in mind that the column count includes
any columns that may ultimately not be printed e.g. those columns defined with NOPRINT or NOZERO.
In the following example SEX is an across variable with WT nested within each distinct value of SEX. With REGION as

the only other variable in the COLUMNS statement, WT appears in _C2_ and _C3_.
Title1 'Mean Weight in Kg';
proc report data=sasclass.clinics nowd;
columns region sex,wt;
define region / group f=$6.;
define sex / across ' patient Sex';
define wt / mean 'Weight' f=6.2;
compute wt;
_c2_ = _c2_/2.2;
_c3_ = _c3_/2.2;
endcomp;
run;
Mean Weight in Kg

region
1
10
2
3
4
5
6
7
8
9

In the compute block
_C2_ refers to the mean
weight of females, while
_C3_ refers to the mean
weight of males. The
order is based on the
ORDER= option which
has a default value of
INTERNAL.

patient Sex
F
M
Weight Weight
.
88.64
33.68
80.45
22.66
47.73
26.40
74.45
29.55
75.27
30.21
80.45
38.64
93.33
.
68.64
72.73
.
36.57
86.59

COMPUTE BLOCK EVENT SEQUENCING
An understanding of the sequencing of events when processing a REPORT step become even more important when the
step includes one or more compute blocks.
When you have multiple compute blocks they can appear in any order in your REPORT step. The order that you include
the blocks does not effect when they are executed. I usually try to group my compute blocks after my BREAK and
RBREAK statements, and I also try to order them nominally in the same order as they will execute. This arrangement
helps me with my coding, but does not change how the compute blocks work.

Report Step Phases
The REPORT step executes in three distinct phases. Having at least a passing understanding of these phases helps the
compute block programmer understand why some things work while others do not.
Evaluation Phase
REPORT begins by evaluating all of the REPORT step statements. If any compute blocks are present the SAS
language elements and LINE statements are set aside for later.
Setup Phase
Next, after the statements have been evaluated, the Setup Phase uses the MEANS/SUMMARY engine to sort the
input data for ORDER and GROUP variables and computes any summarizations. When summarizations or
statistics are calculated, the results are held in the computed summary information which is stored in memory.
Report Row Phase
In this phase, REPORT builds each report row using data from the input data set and/or, when needed, the
computed summary information. If any compute blocks are present they are executed during this phase. REPORT
sends each completed row (one row at a time) to all the ODS destinations (LISTING, PDF, etc.) that are currently
open.

Compute Block Events
As the report is generated (one row at a time - top to bottom and left to right - during the report row phase), the various
compute blocks are executed at the appropriate time. Remember that compute blocks are tied both vertically and
horizontally to locations in the report, and the location controls the timing of when the compute block will be executed.
These ties are specified on the COMPUTE statement. Vertical ties are established by using the timing options BEFORE
and AFTER, and horizontal ties are made by the specification of report items in the COLUMN statement.
When a compute block is tied to a specific report item (like a variable name) on the COLUMN statement, it will be executed
for each report row. The execution will take place when REPORT processes that specific column, and any given row is
processed from left to right based on the order of the items on the COLUMN statement. In the following step, because
SEX is to the left of WEIGHT on the COLUMN statement, the compute block for SEX will be executed before the compute
block for WEIGHT.
proc report .......;
column region sex weight;
.....
compute weight;
.....
endcomp;
compute sex;
.....
endcomp;
.....
run;
Although it would make no difference to the processing of the step, I would reorder the code to place the compute blocks in
the order that they will be processed. The step would become:
proc report .......;
column region sex weight;
.....
compute sex;
.....
endcomp;
compute weight;
.....
endcomp;
.....
run;
Compute blocks that are defined with BEFORE or AFTER on the COMPUTE statement will be executed at the time that
the specified event takes place. When BEFORE or AFTER appears on the COMPUTE statement and there is no report
item, the compute block will be executed only once - BEFORE or AFTER the report.
Including a BEFORE or AFTER on a compute statement is sufficient to generate a row in the computed summary
information during the setup phase. That report row may or may not be written out to the final report depending on the
BREAK and RBREAK specifications. In either case as the report rows are processed during the report row phase, the
compute block is executed as its associated report row is processed.
The COMPUTE BEFORE statement
compute before;
will generate what will be the first row in the computed summary information, while the COMPUTE AFTER statement
compute after;
would generate the last row in the computed summary information and would be the last compute block to execute.

When you have a report item that is a grouping/ordering variable, you can specify the COMPUTE statement with both the
BEFORE/ AFTER and the report item. This allows you to execute the compute block when values of the grouping
variable change. The processing opportunities are roughly similar to FIRST. and LAST. processing in the DATA step. The
following COMPUTE statement sets up a compute block that will be executed just before each new value of REGION.
compute before region;
It is not at all unusual to have compute blocks that execute both BEFORE and AFTER in the same REPORT step.
Compute blocks that execute before the report or group are very useful for initializing variables, while compute blocks that
execute after the report or group are more generally used for summaries.
Using a compute block with a BEFORE or AFTER does not require a corresponding BREAK or RBREAK statement.
However when both are present in the same step the compute block and its corresponding BREAK or RBREAK statement
share a common row in the computed summary information. Remember that the compute block itself does not transfer
information to the report (unless it contains a LINE statement). The compute block can be used to modify information on
the computed summary information, and if there is a corresponding BREAK/RBREAK statement with a SUMMARIZE
option then that information can also be written to the final report.

ANTICIPATING THE REPORT LAYOUT USING OUT=
Visualizing the report layout, especially when ACROSS variables are involved, can be tricky. The OUT= option becomes
more useful as the report becomes more complex.

Exercise #5: Using the OUT= Option
While it is not possible to see the computed summary information directly, you can get an idea as to what it contains by
using the OUT= option.
************************************************************
* Exercise 5;
* Use the OUT= option.;
title1 'Exercise 5';
proc report data=sasclass.clinics
out= outdat nowd;
columns region sex,wt;
define region / group f=$6.;
define sex / across ' patient Sex';
define wt / mean 'Weight' f=6.2;
run;
proc print data=outdat;
run;
The PROC PRINT should give you an idea of what the REPORT table should look like.
Exercise 5
Obs
1
2
3
4
5
6
7
8
9
10

region
1
10
2
3
4
5
6
7
8
9

_C2_
.
163.000
109.667
127.800
143.000
146.200
187.000
.
160.000
177.000

_C3_

_BREAK_

195.000
177.000
105.000
163.800
165.600
177.000
205.333
151.000
.
190.500

The _BREAK_ column becomes very important when using compute blocks across groups.

Exercise #6: Calculate a Ratio using Absolute Column Numbers
Determine the ratio of the mean weight of Females to the mean weight of Males. Hint: Refer back to the result of Exercise
#5.
************************************************************
* Exercise 6;
* Compute the ratio of the mean weight of females
* to the mean weight of males.;
title1 'Exercise 6';
proc report data=sasclass.clinics nowd;
columns region sex,wt ratio;
define region / group f=$6.;
define sex / across ' patient Sex';
define wt / mean 'Weight' f=6.2;
define ratio / computed 'Ratio' f=5.3;
compute ratio;
ratio = xxxxx/yyyyy;
endcomp;
run;
Since SEX has a define usage of ACROSS we will need to use the absolute column numbers to differentiate the values of
the weight of females (_C2_) and males (_C3_). We were able to determine the column numbers by looking at the OUT=
data set (Exercise #5).
title1 'Compute Block Basics';
title2 'Calculating a Ratio';
proc report data=sasclass.clinics nowd;
columns region sex,wt ratio;
define region / group f=$6.;
define sex / across ' patient Sex';
define wt / mean 'Weight' f=6.2;
define ratio / computed 'Ratio' f=5.3;
compute ratio;
ratio = _c2_/ _c3_;
endcomp;
run;

Compute Block Basics
Calculating a Ratio

region
1
10
2
3
4
5
6
7
8
9

patient Sex
F
M
Weight Weight
.
195.00
163.00 177.00
109.67 105.00
127.80 163.80
143.00 165.60
146.20 177.00
187.00 205.33
.
151.00
160.00
.
177.00 190.50

Ratio
.
0.921
1.044
0.780
0.864
0.826
0.911
.
.
0.929

FORMING AND USING GROUPS
The BEFORE and AFTER locations are used with grouping variables to process the rows of tables in groups. Very often
BEFORE is used to initialize within a group while AFTER is used to summarize.
One of the simplest usages is to count items within groups.

Exercise #7: Counting Within Groups
Temporary variables (such as CNT in Exercise #7 below) are used to retain values, but since they do not appear on the
COLUMN statement, they will not appear on the REPORT. Execute the following program to generate a row counter
(COUNT) within each region.
************************************************************
* Exercise 7 ;
* Count items within a group;
* Count patients within REGION;
title1 'Exercise 7';
proc report data=sasclass.clinics nowd;
columns region count lname;
define region / order f=$6.;
define count / computed 'Obs' f=3.;
define lname / display 'Patient';
compute before region;
cnt=0;
endcomp;
compute count;
cnt+1;
count=cnt;
endcomp;
run;
The report item COUNT has the column header ‘Obs’.
Exercise 7
region
1

10

2

Obs
1
2
3
4
1
2
3
4
5
6
1
2
3
4
5
6

Patient
Lawless
Nabers
Mercy
Taber
James
Robertson
Holmes
Manley
Reilly
Hermit
Maxwell
Little
Long
Harbor
Harbor
Haddock

. . . . . portions of the table not shown . . . . .

Calculating Percentages within Groups
Since a percentage calculation is the ratio of the current value to a total for the group, the total must be available for the
calculation. As a result, the calculation of percentages requires the use of two compute blocks. The first is used to
determine the total across the group (COMPUTE BEFORE ì), and then this total is used on each detail row of the table î
to calculate the percentage in the second compute block.
The following example also makes use of the temporary variable TOTWT. A temporary variable does not appear on either
the COLUMN statement or on the report itself. The values of temporary variables are automatically retained in memory as
the rows of the computed summary information are processed, and are therefore very useful for holding onto intermediate
values. Here the variable TOTWT is used to hold the total value of WT.SUM for each group í.
Notice in this example that the compound variable name WT.SUM has two different meanings depending on which
compute block is being executed. When it is used in a summary across all visits to a clinic (ì ð), it will contain the total
weight for that clinic. When a detail row is being processed î, WT.SUM will contain the value of WT for that row.
* Calculating percentages within groups;
title1 'Using the COMPUTE Block';
title2 'Percentages Within Groups';
proc report data=rptdata.clinics nowd split='*';
where clinnum in('031234', '036321');
column clinnum lname wt prctwt ;
define clinnum / group width=10;
define lname
/ order
'Last Name';
define wt
/ analysis format=6.
'Weight';
define prctwt
/ computed format=percent8.1
'Percent*of Total';
compute before clinnum; ì
totwt = wt.sum; í
endcomp;
compute prctwt; î
prctwt = wt.sum / totwt; ï
endcomp;
break after clinnum
run;

/ dol skip summarize suppress; ð

ì This compute block is executed before the individual rows of the table are processed. This makes the temporary
variable (TOTWT) available for use in the second compute block î. The keyword BEFORE is available for this compute
block because CLINNUM is a GROUP variable.
í The total weight is calculated as the total weight within the specific clinic number. Since CLINNUM is a GROUP
variable, WT.SUM will hold the total weight. During the execution of this compute block WT.SUM is the total for the
upcoming CLINNUM, and this value is what is saved in the temporary variable TOTWT.
î This compute block is used to calculate the percentages for each row of the table.
ï In this equation we use the total weight for all the patients that have visited the clinic, TOTWT. This total is used with the
individual patient’s weight (WT.SUM). Eventually the value is displayed using the percent8.1 format.
ð The BREAK statement requests a summary line after each clinic. This statement causes an additional summary row to
be added to the final report and the compute block for PRCTWT will also be executed for the corresponding row in the
computed summary information.

Using the COMPUTE Block
Percentages Within Groups
clinic
number
031234

036321

Last Name
Candle
Henry
Panda
Smith

Weight
195
162
195
162
======
714

Percent
of Total
27.3%
22.7%
27.3%
22.7%
========
100.0% ð

Herbal
Jones
Masters
Stubs

155
105
155
105
======
520

29.8%
20.2%
29.8%
20.2%
========
100.0%

The percentage across the
clinic is of course 100%.
This value is calculated
automatically when the
row resulting from the
BREAK AFTER
CLINNUM statement ð
is processed.
On this row the
temporary variable
TOTWT and WT.SUM
will contain the same
value.

USING THE OUT= OPTION TO VIEW REPORT BREAK INFORMATION
Sometimes when using compute blocks, it is helpful to be able to see the how the break information is being used in the
construction of the report rows. This break information can sometimes be visualized by creating an output data set using
the OUT= option ì on the PROC statement.
After the REPORT step’s computed summary information is processed during the Report Row Phase, the data set that is
specified using the OUT= option is saved as the phase is completed. When this data set is written for us by PROC
REPORT, all the compute block processing has therefore been performed. Although there is really no way to trap or trace
the compute block processing, an examination of the output data set created by PROC REPORT can still be informative.
As a bonus the output data set shows all of the report rows and report items (including those that will not be written to the
final report). In addition to the other report items, the output data set includes the automatic temporary variable _BREAK_.
This variable notes observations in the output data set that are a result of data summarizations, such as those generated
by BREAK and RBREAK statements.
The following example repeats the REPORT step used earlier to demonstrate the calculation of percentages, but it adds
an OUT= option to the PROC REPORT statement.
* Calculating percentages within groups;
title1 'Using the COMPUTE Block';
title2 'Examining the Output Data Set';
proc report data=rptdata.clinics
out=r5_5out ì
nowd split='*';
where clinnum in('031234', '036321');
column clinnum lname wt prctwt ;
define clinnum / group width=10;
define lname
/ order
'Last Name';
define wt
/ analysis format=6.
'Weight';
define prctwt
/ computed format=percent8.1
'Percent*of Total';
compute before clinnum; í
totwt = wt.sum;

endcomp;
compute prctwt;
prctwt = wt.sum / totwt;
endcomp;
break after clinnum
run;

/ dol skip summarize suppress; î

proc print data=r5_5out;
title3 Summarized Data Set;
run;
The PROC PRINT of the data set generated by the OUT= option in this example results in:

Using the COMPUTE Block
Examining the Output Data Set
Summarized Data Set
Obs
1
2
3
4
5
6
7
8
9
10
11
12

clinnum
031234
031234
031234
031234
031234
031234
036321
036321
036321
036321
036321
036321

lname
Candle
Henry
Panda
Smith
Herbal
Jones
Masters
Stubs

wt

prctwt

714
195
162
195
162
714
520
155
105
155
105
520

.
0.27311
0.22689
0.27311
0.22689
1.00000
0.72829
0.29808
0.20192
0.29808
0.20192
1.00000

_BREAK_
clinnum í

clinnum î
clinnum í

clinnum î

í This compute block generates a summary line BEFORE each distinct value of CLINNUM. This row is only a result of
the compute block and does not appear in the final report. Because wt.sum was referenced in a COMPUTE block, we can
see how the value of WT.SUM was captured. Because there is no BREAK statement associated with the COMPUTE
BEFORE, this row does not appear in the final report. However we can see from the output data set, that WT.SUM was
available BEFORE each CLINNUM group in order to be assigned to the temporary variable TOTWT.
î AFTER each group of values for CLINNUM, the BREAK statement generates a summary or break row in the final report,
and this summary row is also reflected in the OUT= data set..
Notice the inclusion of the automatic column _BREAK_. This column can be used to track break events as the table is
calculated. The values of _BREAK_ can be utilized with IF-THEN/ELSE processing in the compute block itself.

Exercise #8: Calculating Percentages
Generate the previous table and examine the table generated by the OUT= option.
************************************************************
* Exercise 8;
* Examine the REPORT step and compare the result
* with the table generated by the OUT= option;
title1 'Exercise 8';
proc report data=sasclass.clinics nowd out=rpttemp;
where clinnum in('031234', '036321');

clinnum
clinnum
lname
wt
prctw

lname wt prctwt ;
/ group width=10;
/ order 'Last Name';
/ analysis format=6. 'Weight';
/ computed format=percent8.1
'Percent/of Total';
compute before clinnum;
totwt = wt.sum;
endcomp;
compute prctwt;
prctwt = wt.sum / totwt;
endcomp;
break after clinnum / dol skip summarize suppress;
run;
proc print data=rpttemp;
run;

COMPUTE BLOCK EXECUTION ORDER
When a compute block is associated with an across variable, the compute block will execute once for each level of the
across variable. This can be avoided by creating a dummy variable.

Exercise #9: Use a Dummy Variable to Control Execution Order
The following exercise converts weight in pounds to kilograms for both males (_C3_) and females (_C2_). Create dummy
variable that is not printed that will do this conversion without executing the compute block twice for each report row.
************************************************************
* Exercise 9;
* Correct the calculations in this step by using a DUMMY
* column;
title1 'Exercise 9';
proc report data=sasclass.clinics nowd;
columns region sex,wt xxxxx;
define region / group f=$6.;
define sex / across ' patient Sex';
define wt / mean 'Weight' f=6.2;
define xxxxx;
compute wt; * fix this line!;
_c2_ = _c2_/2.2;
_c3_ = _c3_/2.2;
endcomp;
run;
Because the variable DUMMY appears to the right of the variable SEX, its compute block will be executed last. Since a
compute block can reference any column to its left, the conversion can take place in the compute block for DUMMY.
title1 'Exercise 9';
proc report data=sasclass.clinics nowd;
columns region sex,wt dummy;
define region / group f=$6.;
define sex / across ' patient Sex';
define wt / mean 'Weight' f=6.2;
define dummy/computed noprint;
compute dummy;
_c2_ = _c2_/2.2;
_c3_ = _c3_/2.2;
endcomp;
run;

SUMMARY
The compute block is unique to the REPORT procedure, and although it adds a great deal of power and flexibility, it also
can add a level of coding complexity that can confound the new user. Fortunately it is not hard to get started using the
compute block. Much of the programmer’s DATA step syntax knowledge can be put to good use in the compute block.
Most SAS language elements can be used in the compute block. In addition the LINE statement can be used to write
blank lines, computed and data values, and text to the report.
When using report items in the compute block, the programmer must be aware of the item’s usage. A report item’s usage
will determine how the item is named or addressed within the compute block.

